


**Médias interactifs numériques**  
**interaction**  
**Processing - 1ers pas**


Pierre Cubaud  
Département d'informatique  
CNAM

Un premier programme pas à pas :



en hommage à Mondrian ...

# primitives graphiques : formes et attributs





- lire la doc en ligne pour chaque fonction de ce programme
- changer l'ordre pour voir l'impact sur le dessin final


```
mondrian1 | Processing 2.0a3
size(400,500);
smooth();


stroke(0);
strokeWeight(10);
strokeCap(SQUARE);

//le fond
background(255);

// le grand carre rouge
fill(255,0,0);
rect(100,0,300,300);
// le rectangle bleu
fill(0,0,255);
rect(0,300,100,200);
// le carre jaune
fill(255,255,0);
rect(300,400,100,100);
// le trait vertical pres du carré jaune
line(300,300,300,400);
// le trait horizontal en haut
line(0,100,100,100);
// le bord
noFill();
rect(0,0,400,500);
|
save("mondrian1.png");
```

## 2ème programme (avec une boucle)





```
size(600,600);
smooth();

background(250);
//noStroke();fill(0,0,128,30);

for (int i=0; i<1000; i++) {

  float x=random(0,600);
  float y=random(0,600);

  ellipse(x,y,60,60);
}

save("generatif1.png");
```

Done Saving.


5




modifier :

- les attributs (stroke, fill)
- le nombre d'itérations
- la taille des cercles

# itération + hasard = génératif !



## 3ème programme : construire une trame



en hommage à Le Parc et Vasarely



```
leparc | Processing 2.0a3  
leparc  
size(600,600);  
rectMode(CENTER);  
ellipseMode(CENTER);  
smooth();  
noStroke();  
  
for (int x=30; x<600; x+=60){  
  for (int y=30; y<600; y+=60){  
    fill(random(0,255));  
    rect(x,y,60,60);  
    fill(random(0,255));  
    ellipse(x,y,50,50);  
  }  
}  
  
save("leparc.png");  
  
Done Saving.  
1
```



modifier :

- la dimension de la trame
- les primitives (triangles ...)
- les couleurs !

# Processing en mode réactif

il faut renseigner deux fonctions prédéfinies :

```
void setup() {  
  // ici du code pour l'initialisation du programme  
}
```

```
void draw() {  
  // là du code exécuté pour chaque trame (frame)  
}
```

# Un premier draw()



démo

en hommage à Vera Molnar

modereactif1

```
void setup() {  
  size(600,600);  
  noFill();stroke(0);  
  background(200);  
  frameRate(10);  
}  
  
void draw() {  
  float x=random(0,width);  
  float y=random(0,height);  
  rect(x,y,50,50);  
}
```



- jouer avec la valeur de frameRate()
- déplacer background() au début de draw()